

NAG Fortran Library Routine Document

F08AHF (SGELQF/DGELQF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08AHF (SGELQF/DGELQF) computes the LQ factorization of a real m by n matrix.

2 Specification

```
SUBROUTINE F08AHF(M, N, A, LDA, TAU, WORK, LWORK, INFO)
ENTRY      sgelqf (M, N, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER    M, N, LDA, LWORK, INFO
real      A(LDA,*), TAU(*), WORK(*)
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine forms the LQ factorization of an arbitrary rectangular real m by n matrix. No pivoting is performed.

If $m \leq n$, the factorization is given by:

$$A = (L \ 0)Q$$

where L is an m by m lower triangular matrix and Q is an n by n orthogonal matrix. It is sometimes more convenient to write the factorization as

$$A = (L \ 0) \begin{pmatrix} Q_1 \\ Q_2 \end{pmatrix}$$

which reduces to

$$A = LQ_1,$$

where Q_1 consists of the first m rows of Q , and Q_2 the remaining $n - m$ rows.

If $m > n$, L is trapezoidal, and the factorization can be written

$$A = \begin{pmatrix} L_1 \\ L_2 \end{pmatrix} Q$$

where L_1 is lower triangular and L_2 is rectangular.

The LQ factorization of A is essentially the same as the QR factorization of A^T , since

$$A = (L \ 0)Q \Leftrightarrow A^T = Q^T \begin{pmatrix} L^T \\ 0 \end{pmatrix}.$$

The matrix Q is not formed explicitly but is represented as a product of $\min(m, n)$ elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 8).

Note also that for any $k < m$, the information returned in the first k rows of the array A represents an LQ factorization of the first k rows of the original matrix A .

4 References

None.

5 Parameters

- 1: M – INTEGER *Input*
On entry: m , the number of rows of the matrix A .
Constraint: $M \geq 0$.

- 2: N – INTEGER *Input*
On entry: n , the number of columns of the matrix A .
Constraint: $N \geq 0$.

- 3: A(LDA,*) – **real** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the m by n matrix A .
On exit: if $m \leq n$, the elements above the diagonal are overwritten by details of the orthogonal matrix Q and the lower triangle is overwritten by the corresponding elements of the m by m lower triangular matrix L .
 If $m > n$, the strictly upper triangular part is overwritten by details of the orthogonal matrix Q and the remaining elements are overwritten by the corresponding elements of the m by n lower trapezoidal matrix L .

- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08AHF (SGELQF/DGELQF) is called.
Constraint: $LDA \geq \max(1, M)$.

- 5: TAU(*) – **real** array *Output*
Note: the dimension of the array TAU must be at least $\max(1, \min(M, N))$.
On exit: further details of the orthogonal matrix Q .

- 6: WORK(*) – **real** array *Workspace*
Note: the dimension of the array $WORK$ must be at least $\max(1, LWORK)$.
On exit: if $INFO = 0$, $WORK(1)$ contains the minimum value of $LWORK$ required for optimum performance.

- 7: LWORK – INTEGER *Input*
On entry: the dimension of the array $WORK$ as declared in the (sub)program from which F08AHF (SGELQF/DGELQF) is called, unless $LWORK = -1$, in which case a workspace query is assumed and the routine only calculates the optimal dimension of $WORK$ (using the formula given below).
Suggested value: for optimum performance $LWORK$ should be at least $M \times nb$, where nb is the **blocksize**.
Constraint: $LWORK \geq \max(1, M)$ or $LWORK = -1$.

- 8: INFO – INTEGER *Output*
On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed factorization is the exact factorization of a nearby matrix $A + E$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*.

8 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}m^2(3n - m)$ if $m \leq n$ or $\frac{2}{3}n^2(3m - n)$ if $m > n$.

To form the orthogonal matrix Q this routine may be followed by a call to F08AJF (SORGLQ/DORGLQ):

```
CALL SORGLQ (N,N,MIN(M,N),A,LDA,TAU,WORK,LWORK,INFO)
```

but note that the first dimension of the array A, specified by the parameter LDA, must be at least N, which may be larger than was required by F08AHF (SGELQF/DGELQF).

When $m \leq n$, it is often only the first m rows of Q that are required, and they may be formed by the call:

```
CALL SORGLQ (M,N,M,A,LDA,TAU,WORK,LWORK,INFO)
```

To apply Q to an arbitrary real rectangular matrix C , this routine may be followed by a call to F08AKF (SORMLQ/DORMLQ). For example,

```
CALL SORMLQ ('Left','Transpose',M,P,MIN(M,N),A,LDA,TAU,C,LDC,
+          WORK,LWORK,INFO)
```

forms the matrix product $C = Q^T C$, where C is m by p .

The complex analogue of this routine is F08AVF (CGELQF/ZGELQF).

9 Example

To find the minimum-norm solutions of the under-determined systems of linear equations

$$Ax_1 = b_1 \text{ and } Ax_2 = b_2$$

where b_1 and b_2 are the columns of the matrix B ,

$$A = \begin{pmatrix} -5.42 & 3.28 & -3.68 & 0.27 & 2.06 & 0.46 \\ -1.65 & -3.40 & -3.20 & -1.03 & -4.06 & -0.01 \\ -0.37 & 2.35 & 1.90 & 4.31 & -1.76 & 1.13 \\ -3.15 & -0.11 & 1.99 & -2.70 & 0.26 & 4.50 \end{pmatrix} \text{ and } B = \begin{pmatrix} -2.87 & -5.23 \\ 1.63 & 0.29 \\ -3.52 & 4.76 \\ 0.45 & -8.41 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08AHF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          MMAX, NMAX, LDA, LDB, NRHMAX, LWORK
```

```

PARAMETER      (MMAX=8,NMAX=8,LDA=MMAX,LDB=NMAX,NRHMAX=NMAX,
+              LWORK=64*NMAX)
real          ZERO, ONE
PARAMETER      (ZERO=0.0e0,ONE=1.0e0)
*
.. Local Scalars ..
INTEGER        I, IFAIL, INFO, J, M, N, NRHS
*
.. Local Arrays ..
real          A(LDA,NMAX), B(LDB,NRHMAX), TAU(NMAX),
+              WORK(LWORK)
*
.. External Subroutines ..
EXTERNAL       sgelqf, sormlq, strsm, F06QHF, X04CAF
*
.. Executable Statements ..
WRITE (NOUT,*) 'F08AHF Example Program Results'
*
Skip heading in data file
READ (NIN,*)
READ (NIN,*) M, N, NRHS
IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.LE.N .AND. NRHS.LE.NRHMAX)
+   THEN
*
*       Read A and B from data file
*
READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,M)
*
*       Compute the LQ factorization of A
*
CALL sgelqf(M,N,A,LDA,TAU,WORK,LWORK,INFO)
*
*       Solve L*Y = B, storing the result in B
*
CALL strsm('Left','Lower','No transpose','Non-Unit',M,NRHS,ONE,
+         A,LDA,B,LDB)
*
*       Set rows (M+1) to N of B to zero
*
IF (M.LT.N) CALL F06QHF('General',N-M,NRHS,ZERO,ZERO,B(M+1,1),
+         LDB)
*
*       Compute minimum-norm solution X = (Q**T)*B in B
*
CALL sormlq('Left','Transpose',N,NRHS,M,A,LDA,TAU,B,LDB,WORK,
+         LWORK,INFO)
*
*       Print minimum-norm solution(s)
*
WRITE (NOUT,*)
IFAIL = 0
*
CALL X04CAF('General',' ',N,NRHS,B,LDB,
+         'Minimum-norm solution(s)',IFAIL)
*
END IF
STOP
END

```

9.2 Program Data

F08AHF Example Program Data

```

4 6 2                                     :Values of M, N and NRHS
-5.42 3.28 -3.68 0.27 2.06 0.46
-1.65 -3.40 -3.20 -1.03 -4.06 -0.01
-0.37 2.35 1.90 4.31 -1.76 1.13
-3.15 -0.11 1.99 -2.70 0.26 4.50      :End of matrix A
-2.87 -5.23
1.63 0.29
-3.52 4.76
0.45 -8.41                             :End of matrix B

```

9.3 Program Results

F08AHF Example Program Results

Minimum-norm solution(s)

	1	2
1	0.2371	0.7383
2	-0.4575	0.0158
3	-0.0085	-0.0161
4	-0.5192	1.0768
5	0.0239	-0.6436
6	-0.0543	-0.6613
